HYDROGEN2SB.MAC, CORRECTED, FROM STEEB & HARDY CH.26

0.1 This script allows to find energy levels of Hydrogen atom from Schroedinger equation, and here we illustrate how to use physical constants in Maxima

```
(%i1) load (physical_constants);
```

Compiling /tmp/gazonk_4257_0.lsp.End of Pass 1. End of Pass 2.

OPTIMIZE levels: Safety=2, Space=3, Speed=3 Finished compiling /tmp/gazonk_4257_0.lsp.

Compiling /tmp/gazonk_4257_0.lsp.End of Pass 1. End of Pass 2.

OPTIMIZE levels: Safety=2, Space=3, Speed=3 Finished compiling /tmp/gazonk_4257_0.lsp.

Compiling /tmp/gazonk_4257_0.lsp.End of Pass 1. End of Pass 2.

OPTIMIZE levels: Safety=2, Space=3, Speed=3 Finished compiling /tmp/gazonk_4257_0.lsp.

Compiling /tmp/gazonk_4257_0.lsp.End of Pass 1. End of Pass 2.

OPTIMIZE levels: Safety=2, Space=3, Speed=3 Finished compiling /tmp/gazonk_4257_0.lsp.

Compiling /tmp/gazonk_4257_0.lsp.End of Pass 1. End of Pass 2.

OPTIMIZE levels: Safety=2, Space=3, Speed=3 Finished compiling /tmp/gazonk_4257_0.lsp.

Compiling /tmp/gazonk_4257_0.lsp.End of Pass 1. End of Pass 2.

OPTIMIZE levels: Safety=2, Space=3, Speed=3Finished compiling /tmp/gazonk_4257_0.lsp.

(%o1) $/usr/share/maxima/5.32.1/share/ezunits/physical\_constants.mac$

Let us use notation from S.Fluegge "Practical Quantum Mechanic", vol.1, problem 61. Kepler's problem for infinetely heavy point-like nucleus of charge Z. For Z=1 we have hydrogen atom. Denote:

```
(%i2) 2*m0*E/hb^2 = -%gamma^2;
```

(%o2) $$\frac{2\,m0\,E}{hb^2} = -\gamma^2$$

here hb denotes $\hbar$, i.e. Planck constant divided by 2π. Let

```
(%i3) kappa = Z*e0^2*m0/(%gamma*hb^2);
```

(%o3) $$\kappa = \frac{e0^2\,m0\,Z}{\gamma\,hb^2}$$

Now for usual substitution in wavefunction $u(r,\theta,\phi)$ to separate variables:

```
(%i4)  u = (1/r) * chi(r) * Ylm(theta,phi);
```

(%o4) $u = \frac{\chi(r)\ \mathrm{Ylm}(\theta,\phi)}{r}$

we write the Schroedinger equation as (with y in place of χ):

```
(%i5)  'diff(y, r, 2)  + (-%gamma^2 + 2*%gamma*kappa/r-l*(l+1)/r^2)* y = 0;
```

(%o5) $\frac{d^2}{d\,r^2}\,y + \left(\frac{2\,\gamma\,\kappa}{r} - \frac{l\,(l+1)}{r^2} - \gamma^2\right)\,y = 0$

Define orbital number l, magnetic m, and radial k, e.g.:

```
(%i6)  l: 1; m: 0; k: 1;
```

(%o6) 1

(%o7) 0

(%o8) 1

now the principal quantum number n is

```
(%i9)  n:k+l+1;
```

(%o9) 3

```
(%i10) alpha: e0^2/(hb*c);
```

(%o10) $\frac{e0^2}{c\,hb}$

alpha is fine structure constant. Now substitute a polynomial $v1$ and exp as an ansatz for y, where

```
(%i11) v1: (sum (a [j] *r^j, j, 1 ,k));
```

(%o11) $a_1\,r$

```
(%i12) v1: v1+1;
```

(%o12) $a_1\,r + 1$

```
(%i13) y1: r^(l+1)*exp(-sqrt(g2)*r)*v1;
```

(%o13) $r^2\,(a_1\,r+1)\,e^{-\sqrt{g2}\,r}$

$y1$ is the trial function of the radial wave function. Insert the ansatz into Schroedinger differential equation

(%i14)
```
left: diff(y1,r,2)+(-g2+2*m0*c*alpha/(hb*r)-l*(l+l)/(r*r))*y1;
left: left*exp(sqrt(g2)*r);
left: left/r;
left: ratsimp(left);
left: num(left);
left: expand(left);
left: subst(x,sqrt(g2),left);
```

(%o14) $-2\,a_1\,\sqrt{g2}\,r^2\,e^{-\sqrt{g2}\,r}+\left(\frac{2\,e0^2\,m0}{hb^2\,r}-\frac{2}{r^2}-g2\right)\,r^2\,(a_1\,r+1)\,e^{-\sqrt{g2}\,r}+ g2\,r^2\,(a_1\,r+1)\,e^{-\sqrt{g2}\,r}-4\,\sqrt{g2}\,r\,(a_1\,r+1)\,e^{-\sqrt{g2}\,r}+2\,(a_1\,r+1)\,e^{-\sqrt{g2}\,r}+4\,a_1\,r\,e^{-\sqrt{g2}\,r}$

(%o15) $\left(-2\,a_1\,\sqrt{g2}\,r^2\,e^{-\sqrt{g2}\,r}+\left(\frac{2\,e0^2\,m0}{hb^2\,r}-\frac{2}{r^2}-g2\right)\,r^2\,(a_1\,r+1)\,e^{-\sqrt{g2}\,r}+g2\,r^2\,(a_1\,r+1)\,e^{-\sqrt{g2}\,r}-4\,\sqrt{}$

(%o16) $\frac{\left(-2\,a_1\,\sqrt{g2}\,r^2\,e^{-\sqrt{g2}\,r}+\left(\frac{2\,e0^2\,m0}{hb^2\,r}-\frac{2}{r^2}-g2\right)\,r^2\,(a_1\,r+1)\,e^{-\sqrt{g2}\,r}+g2\,r^2\,(a_1\,r+1)\,e^{-\sqrt{g2}\,r}-4\,\sqrt{g2}\,r}{r}$

(%o17) $-\frac{\sqrt{g2}\,\left(6\,a_1\,hb^2\,r+4\,hb^2\right)-2\,a_1\,e0^2\,m0\,r-2\,e0^2\,m0-4\,a_1\,hb^2}{hb^2}$

(%o18) $-\sqrt{g2}\,\left(6\,a_1\,hb^2\,r+4\,hb^2\right)+2\,a_1\,e0^2\,m0\,r+2\,e0^2\,m0+4\,a_1\,hb^2$

(%o19) $2\,a_1\,e0^2\,m0\,r-6\,a_1\,\sqrt{g2}\,hb^2\,r+2\,e0^2\,m0-4\,\sqrt{g2}\,hb^2+4\,a_1\,hb^2$

(%o20) $-6\,a_1\,hb^2\,r\,x-4\,hb^2\,x+2\,a_1\,e0^2\,m0\,r+2\,e0^2\,m0+4\,a_1\,hb^2$

(%i21)
```
h[0]: coeff(left,r,0);
h[1]: coeff(left,r,1);
h[2]: coeff(left,r,2);
h[3]: coeff(left,r,3);
h[4]: coeff(left,r,4);
```

(%o21) $-4\,hb^2\,x+2\,e0^2\,m0+4\,a_1\,hb^2$

(%o22) $2\,a_1\,e0^2\,m0-6\,a_1\,hb^2\,x$

(%o23) 0

(%o24) 0

(%o25) 0

(%i26)
```
s: solve ([h[0]=0, h[1]=0, h[2]=0, h[3] =0, h[4]=0],
   [x,a[1],a[2],a[3],a[4]]);
```

(%o26) $[[x=\frac{e0^2\,m0}{2\,hb^2},a_1=0,a_2=\%r1,a_3=\%r2,a_4=\%r3],[x=\frac{e0^2\,m0}{3\,hb^2},a_1=-\frac{e0^2\,m0}{6\,hb^2},a_2=\%r4,a_3=\%r5,a_4=\%r6]]$

```
(%i27) sol: last(s);
```

(%o27) $[x = \frac{e0^2\, m0}{3\, hb^2}, a_1 = -\frac{e0^2\, m0}{6\, hb^2}, a_2 = \%r4, a_3 = \%r5, a_4 = \%r6]$

```
(%i28) sol: first(sol);
```

(%o28) $x = \frac{e0^2\, m0}{3\, hb^2}$

```
(%i29) sol: rhs(sol);
```

(%o29) $\frac{e0^2\, m0}{3\, hb^2}$

```
(%i30) E_kl: -hb^2*sol*sol/2/m0;
```

(%o30) $-\frac{e0^4\, m0}{18\, hb^2}$

energy eigenvalue for k, l

```
(%i31) print ("E_kl=" ,E_kl);
```

$E\_kl = -\frac{e0^4\, m0}{18\, hb^2}$

(%o31) $-\frac{e0^4\, m0}{18\, hb^2}$

The output is the energy eigenvalue E_kl. Now we may use maxima package "physical_constant" which has a set of many dimensional constants. It uses package ezunits which by default employs SI. One can try demo(ezunits); display_known_unit_conversions;

We can list the constants:

```
(%i32) propvars (physical_constant);
```

(%o32) $[\%c, \%mu\_0, \%e\_0, \%Z\_0, \%G, \%h, \%h\_bar, \%m\_P, \%\%k, \%T\_P, \%l\_P, \%t\_P,$
$\%\%e, \%Phi\_0, \%G\_0, \%K\_J,$
$\%R\_K, \%mu\_B, \%mu\_N, \alpha, \%R\_inf, \%a\_0, \%E\_h, \%ratio\_h\_me, \%m\_e, \%N\_A, \%m\_u, \%F, \%R, \%V\_m,$
$\%n\_0, \%ratio\_S0\_R, \sigma, \%c\_1, \%c\_1L, \%c\_2, \%b, \%b\_prime]$

And print some of them, e.g.

```
(%i33) constvalue(%alpha);
```

(%o33) 0.0072973525376

```
(%i34) constvalue (%c);
```

(%o34) $299792458`\frac{m}{s}$

we may copy paste this SI value into CGS value for speed of light (adding two orders):

```
(%i35) c0:29979245800;
```

(%o35) 29979245800

```
(%i36) constvalue (%h_bar);
```

(%o36) $\frac{41412931}{1250000000000000000000000000000000000000000\,\pi}`s\,J$

```
(%i37) float(%);
```

(%o37) $1.054571628251774\,10^{-34}`s\,J$

we may copy paste this SI value into CGS value (7 orders difference in J and erg):

```
(%i38) hb:1.054571628251774e-27;
```

(%o38) $1.054571628251774\,10^{-27}$

```
(%i39) constvalue(%%e);
```

(%o39) $\frac{1602176487}{10000000000000000000000000000}`C$

```
(%i40) float(%);
```

(%o40) $1.602176487\,10^{-19}`C$

```
(%i41) e0:1.602176487e-19*c0/10;
```

(%o41) $4.80320427187535\,10^{-10}$

```
(%i42) constvalue(%m_e);
```

(%o42) $9.10938215\,10^{-31}`kg$

In the same way

```
(%i43) m0:9.10938215e-28;
```

(%o43) $9.10938215\,10^{-28}$

Now we get the energy of the level in ergs

```
(%i44) E: ev(E_kl);
```

(%o44) $-2.4220799671188296\,10^{-12}$

and transform to electron-Volts (do not confuse evaluation operator ev above with eV)

```
(%i45) load("physconst.mac");
```

(%o45) $/usr/share/maxima/5.32.1/share/physics/physconst.mac$

```
(%i46) float(%Ry);
```

(%o46) $\frac{1.0973731568549\,10^{7}}{m}$

Here we get Rydberg in inverse meters. Next line gives its uncertainty:

```
(%i47)  get(%Ry, RSU);
```

(%o47) $7.6\,10^{-12}$

We better get Ry in eV. We must multiply electron charge e0 by one Volt, expressed in CGS. The unit of voltage in CGS is 1 statvolt = 299.792458 volts. (The conversion factor 299.792458 is simply the numerical value of the speed of light in cm/s divided by 10^8). Hence,

```
(%i48) eV0:e0/(c0*1e-8);
```

(%o48) $1.602176487\,10^{-12}$

This is the value of eV in ergs, now we have the energy of our level in eV:

```
(%i49) E_eV:E/eV0;
```

(%o49) -1.511743548086928

Since Rydberg is

```
(%i50) Ry0:m0*e0^4/hb^2;
```

(%o50) $4.3597439408138926\,10^{-11}$

in ergs, and in eV

```
(%i51) Ry0eV:Ry0/eV0;
```

(%o51) 27.2113838655647

We obtain the energy of our level in Rydbergs:

```
(%i52) E/Ry0;
```

(%o52) -0.055555555555555

It is easy to check that we get $-1/(2n^2)$ when the principal quantum number of the level is n.